

# INFOAFP – Exam

Andres Löh

Wednesday, 16 April 2008, 09:00–12:00

## Solutions

- Not all possible solutions are given.
- In many places, much less detail than I have provided in the example solution was actually required.
- Solutions may contain typos.

## Evaluation strategies (19 points total)

1 (5 points). Give an example of a Haskell expression of type *Bool* that evaluates to *True* and that would not terminate (i.e., loop forever) in a language with strict evaluation.

Using equational reasoning, give the reduction sequence of your expression to *True* and indicate clearly where a strict reduction strategy would select another redex. •

*Solution 1.* Here is one option:

$$example_1 = head (repeat True)$$

And the reduction:

$$\begin{aligned} example_1 &\equiv \{ \text{definition of } example_1 \} \\ &\quad head (repeat True) \\ &\equiv \{ \text{definition of } repeat \} \\ &\quad head (True : repeat True) \\ &\equiv \{ \text{definition of } head \text{ (strict evaluation would reduce } repeat \text{ True again)} \} \\ &\quad True \end{aligned}$$

Another option:

$$\begin{aligned} loop &= loop \\ example_2 &= const True loop \end{aligned}$$

And the reduction:

$$\begin{aligned} example_2 &\equiv \{ \text{definition of } example_2 \} \\ &\quad const True loop \\ &\equiv \{ \text{definition of } const \text{ (strict evaluation would reduce } loop) \} \\ &\quad True \end{aligned}$$

○

2 (4 points). Haskell has non-strict semantics (i.e., the implementations use lazy evaluation) and is called a pure language. (S)ML on the other hand has strict semantics and is an impure language.

What does purity mean in this context? Give an (small!) example of why Haskell is considered pure and (S)ML is not. [Syntactic correctness, particularly of (S)ML code, is not important.] •

*Solution 2.* Purity refers to the fact that functions are pure. A pure function depends only on its input, i.e., will always have the same result when applied to a specific input. In SML, one can write

**val** *x* = **ref** 0

and then

```
fun f () = (x := !x + 1; !x)
```

The function  $f$  has type  $unit \rightarrow int$ , i.e., there is only one value  $()$  we can apply  $f$  to. Consequently, if pure,  $f$  would have to be a constant function – but  $f$  will produce different results because it depends on the current value of the reference  $x$  as a side effect.

In Haskell, all functions (except for corner cases such as *unsafePerformIO*) are pure. Side effects are only possible by producing values of certain built-in types such as *IO* that are subsequently interpreted by the run-time system.  $\circ$

**3** (4 points). Would a lazy impure language or a strict pure language be possible? Would such programming languages be useful? Discuss briefly.  $\bullet$

*Solution 3.* Both variants are possible. While some impure languages are known to allow laziness in selected areas, a generally lazy impure language would probably not be very useful, because effects and laziness do not mix well. It becomes extremely hard to track when side effects will occur.

A strict pure language can be useful. Strict languages do not have  $\perp$  as a value of every datatype, and that makes reasoning about programs simpler. Most strict languages are impure just because strict evaluation is easier to track for the programmer, and impure handling of side effects may seem more convenient. There is, however, no good reason not to be more disciplined in a strict setting.  $\Omega$ mega is an example of a strict pure language.  $\circ$

**4** (6 points). Consider the following Haskell functions. Divide the functions into equivalence classes, i.e., group the functions that are semantically equivalent (efficiency is irrelevant). Give as many examples as needed to demonstrate that each of the classes has indeed different behaviour.

```
s1 :: [a] → b → b
s1 xs y = case xs of { [] → y; _ → y }

s2 :: [a] → b → b
s2 xs y = seq xs y

s3 :: [a] → b → b
s3 xs y = y

s4 :: [a] → b → b
s4 xs y = if null xs then y else y

s5 :: [a] → b → b
s5 xs y = if map (const 0) xs == [] then y else y

s6 :: [a] → b → b
```

$s6\ xs\ y = \text{case } xs \text{ of } \{ [] \rightarrow y; [x] \rightarrow y; _ \rightarrow y \}$

$s7 :: [a] \rightarrow b \rightarrow b$

$s7\ xs\ y = seq\ [xs]\ y$

•

*Solution 4.* There are three equivalence classes that can be distinguished by looking at the cases  $xs = \perp$  and  $xs = 0 : \perp$ .

Functions  $s3$  and  $s7$  return  $y$  in all cases.

Functions  $s1$ ,  $s2$ ,  $s4$  and  $s5$  evaluate the first argument to weak head normal form, and therefore return  $\perp$  if  $xs = \perp$ , but return  $y$  for  $xs = 0 : \perp$  because that is already in weak head normal form.

Function  $s6$  has a pattern  $[x]$  and therefore has to evaluate the tail of  $xs$  to weak head normal form as well. It will therefore return  $\perp$  for both  $xs = \perp$  and  $xs = 0 : \perp$ . ◦

## Interactive programs (12 points total)

Consider the following datatype:

```
data GP a = End a
          | Get (Int → GP a)
          | Put Int (GP a)
```

A value of type  $GP$  can be used to describe programs that read and write integer values and return a final result of type  $a$ . Such a program can end immediately (*End*). If it reads an integer, the rest of the program is described as a function depending on this integer (*Get*). If the program writes an integer (*Put*), the value of that integer and the rest of the program are recorded.

The following expression describes a program that continuously reads integers and prints them:

$echo = Get (\lambda n \rightarrow Put\ n\ echo)$

5 (1 point). What is the (inferred) type of  $echo$ ? •

*Solution 5.*

$echo :: GP\ a$

◦

6 (4 points). Write a function

$run :: GP\ a \rightarrow IO\ a$

that can run a  $GP$ -program in the  $IO$  monad. A *Get* should read an integer from the console, and *Put* should write an integer to the console.

Here is an example run from GHCi:

```
Main> run echo
? 42
42
? 28
28
? 1
1
? - 5
- 5
? Interrupted.
Main>
```

[To better distinguish inputs from outputs, this version of *run* prints a question mark when expecting an input. It is not required that your version does the same.] •

*Solution 6.* Here is the variant which prints the question mark. For this to run properly in a compiled program, we would also have to flush *stdout* or modify the buffering behaviour of *stdout*.

```
run (End x) = return x
run (Get g) = putStr "? " >> getLine >>= run ∘ g ∘ read
run (Put n k) = putStrLn (show n) >> run k
```

Simple version without prompt:

```
run' (End x) = return x
run' (Get g) = getLine >>= run' ∘ g ∘ read
run' (Put n k) = putStrLn (show n) >> run' k
```

It would of course be nice to also handle incorrect inputs more gracefully than *read* does. ○

7 (3 points). Write a GP-program *add* that reads two integers, writes the sum of the two integers, and ultimately returns (). •

*Solution 7.*

```
add :: GP ()
add = Get (λx → Get (λy → Put (x + y) (End ())))
```

○

8 (4 points). Write a GP-program *accum* that reads an integer. If the integer is 0, it returns the current total. If the integer is not 0, it adds the integer to the current total, prints the current total, and starts from the beginning. •

Solution 8.

```

accum :: GP Int
accum = accum' 0

accum' :: Int → GP Int
accum' t = Get (λn → if n == 0 then End t
                    else Put (t + n) (accum' (t + n)))

```

○

## Simulation (21 points total)

9 (4 points). Instead of running a GP-program in the IO monad, we can also simulate the behaviour of such a program by providing a (possibly infinite) list of input values. Write a function

```

simulate :: GP a → [Int] → (a, [Int])

```

that takes such a list of input values and returns the final result plus the (possibly infinite) list of all the output values generated. ●

Solution 9. This is the straight-forward solution:

```

simulate (End x) inp      = (x, [])
simulate (Get g) (i : inp) = simulate (g i) inp
simulate (Put n k) inp     = let (r, out) = simulate k inp
                               in (r, n : out)

```

Using an accumulating argument to hold the list of outputs is not helpful, because then it is not possible to analyze partial results – in particular, *echo* can never be simulated with an accumulating version because it never ends. ○

10 (3 points). What is the result of evaluating the following two expressions?

```

simulate accum [5,4..0]

```

```

simulate accum [5,4..1]

```

●

Solution 10. The first expression results in

```

(15, [5, 9, 12, 14, 15])

```

whereas the second expression results in a pattern match failure. Note that the exact nature of the result of the second expression depends on how *simulate* deals with an inputs list that is not sufficiently long. ○

11 (4 points). Define a QuickCheck property that states the following property using *simulate*:

“If *echo* is given *n* numbers as input, then the first *n* numbers of its output will be identical to the input.” •

Solution 11.

```
echoP :: [Int] → Bool
echoP xs = take (length xs) (snd (simulate echo xs)) == xs
```

Or even simpler:

```
echoP' :: [Int] → Bool
echoP' xs = xs `isPrefixOf` snd (simulate echo xs)
```

using *isPrefixOf* from *Data.List*. Indeed, these properties are fulfilled – evaluating

```
tests = quickCheck echoP >> quickCheck echoP'
```

results in

```
+++ OK, passed 100 tests.
+++ OK, passed 100 tests.
```

Using an accumulating version of *simulate*, the properties will fail immediately. ○

12 (4 points). Which parts of the definition of *simulate* are covered by your property, and which are not? (I.e., which parts of the definition of *simulate* would be highlighted by HPC after running QuickCheck on your property – assuming that QuickCheck generates suitably random lists.) •

Solution 12.

Here is the actual output of HPC:

```
31
32 simulate (End x) inp = (x, [])
33 simulate (Get g) (i:inp) = simulate (g i) inp
34 simulate (Put n k) inp = let (r, out) = simulate k inp
35 in (r, n:out)
36
```

The rhs of the case for *End* is not covered, because *echo* does not ever produce *End*. Also, the first component of the resulting pair is not covered, because the property only checks the list, i.e., the second component.

Using an accumulating version of *simulate*, the *End* case will also not be covered, plus the accumulator itself will not be covered in all the cases – since the property always fails with an exception, there will never be a chance to actually perform the comparison on the resulting list. ○

13 (6 points). This is an attempt to define a QuickCheck property for *accum*:

```
accumP :: [Int] → Property
accumP xs = all (λx → x > 0) xs ==>
    simulate accum (xs ++ [0]) == (last sl, sl)
    where sl = scanl1 (+) xs
```

Here, *scanl1* is defined as follows

```
scanl1 :: (a → a → a) → [a] → [a]
scanl1 f [] = []
scanl1 f (x : xs) = scanl f x xs

scanl :: (a → b → a) → a → [b] → [a]
scanl f x xs = x : case xs of
    [] → []
    y : ys → scanl f (f x y) ys
```

There are at least two problems with this property. Describe how they can be fixed [a description is sufficient]. •

*Solution 13.* The definition of *accumP* fails with an exception on an empty input list due to the call to *last*. There are several ways to fix this. For instance, one can use a variant of *last* that returns 0 on the empty list.

Also, the condition that all values of the list must be positive will quickly exhaust the test values. One can use a custom generator to fix this problem.

Here is a new version of the property that works:

```
accumP' :: Property
accumP' = forAll pos $ λxs →
    let sl = scanl1 (+) xs
    in simulate accum (xs ++ [0]) == (last' sl, sl)

last' xs = if null xs then 0 else last xs

pos :: Gen [Int]
pos = sized $ λn → do
    k ← choose (0, n)
    replicateM k (choose (1, n + 1))
```

Calling *quickCheck accumP'* yields:

```
+++ OK, passed 100 tests.
```

○



## Functors and monads (24 points total)

A map function for  $GP$  can be defined as follows:

```
instance Functor GP where
  fmap f (End x)  = End (f x)
  fmap f (Get g)  = Get (fmap f ∘ g)
  fmap f (Put n x) = Put n (fmap f x)
```

**14** (2 points). Describe the difference between the behaviour of *run accum* and the behaviour of *run (fmap (\*2) accum)*. •

*Solution 14.* Only the final result will be doubled. The values printed during the execution of *accum* are unchanged. ○

Instances of class *Functor* should generally fulfill the following two laws:

$$\begin{aligned} \forall x. \quad \text{fmap id } x &\equiv x \\ \forall f \ g \ x. \quad \text{fmap } (f \circ g) \ x &\equiv \text{fmap } f \ (\text{fmap } g \ x) \end{aligned}$$

**15** (8 points). Prove the **first** of the two laws using equational reasoning (and ignoring that values can be  $\perp$ ).

Note that if you want to prove a property  $P \ p$  for any  $p :: GP \ a$  via structural induction, you have to prove the following three cases:

$$\begin{aligned} \forall x. \quad P \ (\text{End } x) \\ \forall g. \quad (\forall x. P \ (g \ x)) &\Rightarrow P \ (\text{Get } g) \\ \forall n \ p. \quad P \ p &\Rightarrow P \ (\text{Put } n \ p) \end{aligned}$$

(Here,  $\Rightarrow$  denotes logical implication.) Note that the second case is slightly unusual due to the function argument of *Get*: you may assume that  $P \ (g \ x)$  holds for any value of  $x$ ! •

*Solution 15.* We prove  $\forall y. \text{fmap id } y \equiv y$  by structural induction on  $y :: GP \ a$ .

Case  $y = \text{End } x$ :

$$\begin{aligned} &\text{fmap id } (\text{End } x) \\ \equiv &\quad \{ \text{definition of fmap} \} \\ &\text{End } (\text{id } x) \\ \equiv &\quad \{ \text{definition of id} \} \\ &\text{End } x \end{aligned}$$

Case  $y = \text{Get } g$ :

$$\begin{aligned} &\text{fmap id } (\text{Get } g) \\ \equiv &\quad \{ \text{definition of fmap} \} \\ &\text{Get } (\text{fmap id } \circ g) \end{aligned}$$

$\equiv$  { definition of  $(\circ)$  }  
 $\text{Get } (\lambda x \rightarrow \text{fmap id } (g x))$   
 $\equiv$  { induction hypothesis }  
 $\text{Get } (\lambda x \rightarrow g x)$   
 $\equiv$  { eta-reduction (i.e., rewriting to point-free form) }  
 $\text{Get } g$

Case  $y = \text{Put } n x$ :

$\text{fmap id } (\text{Put } n x)$   
 $\equiv$  { definition of  $\text{fmap}$  }  
 $\text{Put } n (\text{fmap id } x)$   
 $\equiv$  { induction hypothesis }  
 $\text{Put } n x$

○

**16 (5 points).** Define a sensible monad instance for *GP*. •

*Solution 16.*

**instance Monad GP where**  
 $\text{return } x = \text{End } x$   
 $(\text{Get } g) \gg= f = \text{Get } (\lambda n \rightarrow g n \gg= f)$   
 $(\text{Put } n x) \gg= f = \text{Put } n (x \gg= f)$   
 $(\text{End } x) \gg= f = f x$

○

**17 (5 points).** Define a sensible *MonadState* instance for *GP*. Recall the *MonadState* class:

**class (Monad m)  $\Rightarrow$  MonadState s m | m  $\rightarrow$  s where**  
 $\text{get} :: m s$   
 $\text{put} :: s \rightarrow m ()$

•

*Solution 17.*

**instance MonadState Int GP where**  
 $\text{get} = \text{Get } (\lambda n \rightarrow \text{End } n)$   
 $\text{put } n = \text{Put } n (\text{End } ())$

○

**18 (4 points).** What is the difference between the normal state monad as defined in module *Control.Monad.State* and *GP*? Discuss whether you think it is a good idea to make *GP* an instance of *MonadState*. •

*Solution 18.* While *State* maintains a single piece of state that can be affected using *get* and *put*, *GP* makes use of two independent streams for input and output. As a consequence, two subsequent invocations of *get* can produce different results, or a *get* immediately after *put* will normally produce a different value than the one written using *put*. In other words, properties one might expect for instances of *MonadState* such as

$$\begin{aligned} \forall f x. \quad \mathbf{do} \{ \mathit{get}; x \leftarrow \mathit{get}; f x \} &\equiv \mathbf{do} \{ x \leftarrow \mathit{get}; f x \} \\ \forall f x y. \quad \mathbf{do} \{ \mathit{put} x; y \leftarrow \mathit{get}; f y \} &\equiv \mathbf{do} \{ \mathit{put} x; f x \} \end{aligned}$$

do not hold for *GP*. On the positive side, it is not given that all instances of *MonadState* should have these properties, and *GP* does implement the *MonadState* interface – so one can argue that it should be reused.  $\circ$

## Type classes (10 points total, 5 bonus points)

19 (2 points). Consider this program:

```
equal :: (Eq s, MonadState s m) => m Bool
equal = do
    x <- get
    y <- get
    return (x == y)
```

Is the given type signature the most general type signature for *equal*? What would happen if the type signature would be omitted?  $\bullet$

*Solution 19.* Yes, the type signature gives the most general type. Without the type signature, one would get a type error due to the monomorphism restriction. If the monomorphism restriction is explicitly disabled in GHC, then GHC is able to infer the type.  $\circ$

20 (8 points). Translate type classes into explicit evidence in the above function *equal*. Desugar the **do**-notation in the process [use the “simple” desugaring, without the possibility to pattern match on the left hand side of an arrow]. Define the dictionary types that are required – you may omit class methods that are not relevant to this example. You may also declare local abbreviations using **let**.  $\bullet$

*Solution 20.*

```
data EqD a          = EqD      { eq :: a -> a -> Bool }

data MonadD m       = MonadD   { ret :: forall a. a -> m a,
                                bind :: forall a b. m a -> (a -> m b) -> m b }

data MonadStateD s m = MonadStateD { monad :: MonadD m,
                                       gt :: m s }
```

```

equalD :: EqD s → MonadStateD s m → m Bool
equalD eqD msD =
  let (≫=) = bind (monad msD)
      return = ret (monad msD)
  in  gt msD ≻= λx →
      gt msD ≻= λy →
      return (eq eqD x y)

```

○

**21 (5 bonus points).** Haskell does not offer a scoping mechanism for instances. Instances are always exported from modules, even if nothing else is. Also, instances cannot be local. For example,

```

let instance Eq Char where
    x == y = ord (toUpper x) == ord (toUpper y)
in  "hello" == "HeLlo"

```

(using `ord` and `toUpper` from `Data.Char`) is not legal Haskell.

Why do you think this decision has been made? Are there any problems you can think of? ●

*Solution 21.* The main problem is that it becomes tricky to track when the local instances apply. The reason is that context reduction does not always take place immediately.

In the example above, one would probably expect that the type of the expression is `Bool` and the local instance is used. However, what if we replace the expression after `in` by

```
λxs ys → xs ++ ys == ys ++ xs
```

The type of this function is  $Eq\ a \Rightarrow [a] \rightarrow [a] \rightarrow Bool$ . At this point, it is not yet clear if the function will be applied to strings or to some other lists.

One can of course specify the behaviour of this and comparable cases, but it is not immediately clear which behaviour would be most intuitive, and all such changes would at least make the language more complicated.

Exactly the same problems apply to instances that are local to a module. ○

## GADTs and kinds (14 points total)

Here is a variation of *GP*:

```

data GP' :: * → * where
  Return :: a → GP' a
  Bind    :: GP' a → (a → GP' b) → GP' b
  Get'    :: GP' Int
  Put'    :: Int → GP' ()

```

This is a GADT. The type  $GP'$  can trivially be made an instance of the classes *Monad* and *MonadState*:

**instance** *Monad*  $GP'$  **where**

*return* = *Return*

$(\gg=)$  = *Bind*

**instance** *MonadState* *Int*  $GP'$  **where**

*get* = *Get'*

*put* = *Put'*

**22** (6 points). A value of type  $GP$  can easily be transformed into a value of type  $GP'$  as follows:

$gp2gp' :: GP\ a \rightarrow GP'\ a$   
 $gp2gp' (End\ x) = Return\ x$   
 $gp2gp' (Get\ f) = Get' \gg= \lambda x \rightarrow gp2gp' (f\ x)$   
 $gp2gp' (Put\ n\ k) = Put'\ n \gg gp2gp' k$

Define a transformation in the other direction, i.e., a function

$gp'2gp :: GP'\ a \rightarrow GP\ a$

such that  $gp'2gp \circ gp2gp' \equiv id$  for all values that do not contain  $\perp$ . Does  $gp2gp' \circ gp'2gp$  also yield the identity? [No formal proof is required.] •

*Solution 22.* In this version, the already defined monadic structure on  $GP$  is used. It is, of course, possible to expand the definitions of *return*,  $(\gg=)$ , *get* and *put*.

$gp'2gp (Return\ x) = return\ x$   
 $gp'2gp (Bind\ m\ f) = gp'2gp\ m \gg= \lambda x \rightarrow gp'2gp (f\ x)$   
 $gp'2gp\ Get' = get$   
 $gp'2gp (Put'\ x) = put\ x$

The composition  $gp2gp' \circ gp'2gp$  is not the identity. As an example, consider the expression  $Bind\ (Return\ 0)\ Put' :: GP'\ ()$ . A formal proof was not required, but here it is anyway:

$(gp2gp' \circ gp'2gp) (Bind\ (Return\ 0)\ Put')$   
 $\equiv \{ \text{definition of } (\circ) \}$   
 $gp2gp' (gp'2gp (Bind\ (Return\ 0)\ Put'))$   
 $\equiv \{ \text{definition of } gp'2gp \}$   
 $gp2gp' (gp'2gp (Return\ 0) \gg= \lambda x \rightarrow gp'2gp (Put'\ x))$   
 $\equiv \{ \text{definition of } gp'2gp, \text{ twice} \}$   
 $gp2gp' (return\ 0 \gg= \lambda x \rightarrow put\ x)$   
 $\equiv \{ \text{definition of } return \text{ for } GP \}$

$$\begin{aligned}
& gp2gp' (End\ 0 \gg \lambda x \rightarrow put\ x) \\
\equiv & \{ \text{definition of } (\gg) \text{ for } GP \} \\
& gp2gp' ((\lambda x \rightarrow put\ x)\ 0) \\
\equiv & \{ \text{beta-reduction} \} \\
& gp2gp' (put\ 0) \\
\equiv & \{ \text{definition of } put \text{ for } GP \} \\
& gp2gp' (Put\ 0\ (End\ ())) \\
\equiv & \{ \text{definition of } gp2gp' \} \\
& Put'\ 0 \gg gp2gp' (End\ ()) \\
\equiv & \{ \text{definition of } (\gg) \} \\
& Put'\ 0 \gg \lambda _ \rightarrow gp2gp' (End\ ()) \\
\equiv & \{ \text{definition of } (\gg) \text{ for } GP' \} \\
& Bind\ (Put'\ 0)\ (\lambda _ \rightarrow gp2gp' (End\ ())) \\
\equiv & \{ \text{definition of } gp2gp' \} \\
& Bind\ (Put'\ 0)\ (\lambda _ \rightarrow Return\ ())
\end{aligned}$$

The result  $Bind\ (Put'\ 0)\ (\lambda _ \rightarrow Return\ ())$  is not equal to  $Bind\ (Return\ 0)\ Put' :: GP' ()$ . ○

**23** (4 points). Do the monad laws hold for  $GP$  and  $GP'$ ? [Give a counterexample if not, argue briefly if yes – no formal proof is required.]

Describe advantages and disadvantages of the two variants. ●

*Solution 23.* It is easy to see that  $GP'$  does not fulfill the monad laws. The monad laws require that

$$return\ 0 \gg put \equiv put\ 0$$

However,  $Bind\ (Return\ 0)\ Put'$  is not equal to  $Put'\ 0$ .

On the other hand,  $GP$  fulfills the monad laws. Here is a proof for completeness (even though it was not required):

We first show that

$$\forall x\ f. \quad return\ x \gg f \equiv f\ x$$

We can prove this directly, without induction:

$$\begin{aligned}
& return\ x \gg f \\
\equiv & \{ \text{definition of } return \text{ for } GP \} \\
& End\ x \gg f \\
\equiv & \{ \text{definition of } (\gg) \text{ for } GP \} \\
& f\ x
\end{aligned}$$

The next law requires that

$$\forall m. \quad m \gg return \equiv m$$

We prove this by structural induction on  $m$ :

Case  $m = \text{End } x$ :

$$\begin{aligned}
& \text{End } x \gg= \text{return} \\
& \equiv \{ \text{definition of } (\gg=) \text{ for GP} \} \\
& \quad \text{return } x \\
& \equiv \{ \text{definition of } \text{return} \text{ for GP} \} \\
& \quad \text{End } x
\end{aligned}$$

Case  $m = \text{Get } g$ :

$$\begin{aligned}
& \text{Get } g \gg= \text{return} \\
& \equiv \{ \text{definition of } (\gg=) \text{ for GP} \} \\
& \quad \text{Get } (\lambda n \rightarrow g \ n \gg= \text{return}) \\
& \equiv \{ \text{induction hypothesis} \} \\
& \quad \text{Get } (\lambda n \rightarrow g \ n) \\
& \equiv \{ \text{eta-reduction, i.e., rewrite to point-free form} \} \\
& \quad \text{Get } g
\end{aligned}$$

Case  $m = \text{Put } n \ x$ :

$$\begin{aligned}
& \text{Put } n \ x \gg= \text{return} \\
& \equiv \{ \text{definition of } (\gg=) \text{ for GP} \} \\
& \quad \text{Put } n \ (x \gg= \text{return}) \\
& \equiv \{ \text{induction hypothesis} \} \\
& \quad \text{Put } n \ x
\end{aligned}$$

As final step, we have to show the associativity of  $(\gg=)$ , i.e., that

$$\forall m \ f \ h. \ (m \gg= f) \gg= h \equiv m \gg= (\lambda x \rightarrow f \ x \gg= h)$$

We show this by induction on  $m$ .

Case  $m = \text{End } x$ :

$$\begin{aligned}
& (\text{End } x \gg= f) \gg= h \\
& \equiv \{ \text{definition of } (\gg=) \text{ for GP} \} \\
& \quad f \ x \gg= h \\
& \equiv \{ \text{abstracting from } x \} \\
& \quad (\lambda x \rightarrow f \ x \gg= h) \ x \\
& \equiv \{ \text{definition of } (\gg=) \text{ for GP} \} \\
& \quad \text{End } x \gg= (\lambda x \rightarrow f \ x \gg= h)
\end{aligned}$$

Case  $m = \text{Get } g$ :

$$\begin{aligned}
& (\text{Get } g \gg= f) \gg= h \\
& \equiv \{ \text{definition of } (\gg=) \text{ for GP} \}
\end{aligned}$$

$$\begin{aligned}
& \text{Get } (\lambda n \rightarrow g \ n \gg= f) \gg= h \\
& \equiv \{ \text{definition of } (\gg=) \text{ for } GP \} \\
& \text{Get } (\lambda m \rightarrow (\lambda n \rightarrow g \ n \gg= f) \ m \gg= h) \\
& \equiv \{ \text{beta-reduction} \} \\
& \text{Get } (\lambda m \rightarrow (g \ m \gg= f) \gg= h) \\
& \equiv \{ \text{induction hypothesis} \} \\
& \text{Get } (\lambda m \rightarrow g \ m \gg= (\lambda x \rightarrow f \ x \gg= h)) \\
& \equiv \{ \text{definition of } (\gg=) \text{ for } GP \} \\
& \text{Get } g \gg= (\lambda x \rightarrow f \ x \gg= h)
\end{aligned}$$

Case  $m = \text{Put } n \ k$ :

$$\begin{aligned}
& (\text{Put } n \ k \gg= f) \gg= h \\
& \equiv \{ \text{definition of } (\gg=) \text{ for } GP \} \\
& \text{Put } n \ (k \gg= f) \gg= h \\
& \equiv \{ \text{definition of } (\gg=) \text{ for } GP \} \\
& \text{Put } n \ ((k \gg= f) \gg= h) \\
& \equiv \{ \text{induction hypothesis} \} \\
& \text{Put } n \ (k \gg= (\lambda x \rightarrow f \ x \gg= h)) \\
& \equiv \{ \text{definition of } (\gg=) \text{ for } GP \} \\
& \text{Put } n \ k \gg= (\lambda x \rightarrow f \ x \gg= h)
\end{aligned}$$

This completes the proof of the monad laws for  $GP$ . ○

**24** (4 points). Define type synonyms of kind

$$((\ast \rightarrow \ast) \rightarrow \ast) \rightarrow \ast$$

and

$$(\ast \rightarrow \ast) \rightarrow (\ast \rightarrow \ast) \rightarrow (\ast \rightarrow \ast)$$

without using any user-defined **datatypes**. ●

*Solution 24.*

```

type F f = f []
type G f g x = (f x, g x)

```

```

Main⟩ : kind F
F :: ((∗ → ∗) → ∗) → ∗
Main⟩ : kind G
G :: (∗ → ∗) → (∗ → ∗) → ∗ → ∗

```

○