

INFOAFP – Exam

Andres Löh

Monday, 19 April 2010, 09:00–12:00

Solutions

- Not all possible solutions are given.
- In many places, much less detail than I have provided in the example solution was actually required.
- Solutions may contain typos.

Zippers (33 points total)

A *zipper* is a data structure that allows navigation in another tree-like structure. Consider binary trees:

```
data Tree a = Leaf a | Node (Tree a) (Tree a)
deriving (Eq, Show)
```

A *one-hole context* for trees is given by the following datatype:

```
data TreeCtx a = NodeL () (Tree a) | NodeR (Tree a) ()
deriving (Eq, Show)
```

The idea is as follows: leaves contain no subtrees, therefore they do not occur in the context type. In a node, we can focus on either the left or the right subtree. The context then consists of the other subtree. The use of `()` is just to mark the position of the hole – it is not really needed.

We can plug a tree into the hole of a context as follows:

```
plugTree :: Tree a → TreeCtx a → Tree a
plugTree l (NodeL () r) = Node l r
plugTree r (NodeR l ()) = Node l r
```

A zipper for trees encodes a tree where a certain subtree is currently in focus. Since the focused tree can be located deep in the full tree, one element of type `TreeCtx a` is not sufficient. Instead, we store the focused subtree together with a *list* of one-layer contexts that encodes the path from the focus to the root node:

```
data TreeZipper a = TZ (Tree a) [TreeCtx a]
deriving (Eq, Show)
```

We can recover the full tree from the zipper as follows:

```
leave :: TreeZipper a → Tree a
leave (TZ t cs) = foldl plugTree t cs
```

Consider the tree

```
tree :: Tree Char
tree = Node (Node (Leaf 'a') (Leaf 'b'))
           (Node (Leaf 'c') (Leaf 'd'))
```

If we focus on the rightmost leaf containing `'d'`, the corresponding zipper structure is

```
example :: TreeZipper Char
example = TZ (Leaf 'd')
            [NodeR (Leaf 'c') (), NodeR (Node (Leaf 'a') (Leaf 'b')) ()]
```

1 (3 points). Define a function

$enter :: Tree\ a \rightarrow TreeZipper\ a$

that creates a zipper from a tree such that the full tree is in focus. •

Solution 1.

$enter\ t = TZ\ t\ []$

Moving the focus from a tree down to the left subtree works as follows: ○

$down :: TreeZipper\ a \rightarrow Maybe\ (TreeZipper\ a)$
 $down\ (TZ\ (Leaf\ x)\ cs) = Nothing$
 $down\ (TZ\ (Node\ l\ r)\ cs) = Just\ (TZ\ l\ (NodeL\ ()\ r\ : cs))$

The function fails if there is no left subtree, i. e., if we are in a leaf.

2 (8 points). Define functions

$up :: TreeZipper\ a \rightarrow Maybe\ (TreeZipper\ a)$
 $right :: TreeZipper\ a \rightarrow Maybe\ (TreeZipper\ a)$

that move the focus from a subtree to its parent node or to its right sibling, respectively. Both functions should fail (by returning *Nothing*) if the move is not possible. •

Solution 2. These are the simple definitions:

$up\ (TZ\ t\ (c : cs)) = Just\ (TZ\ (plugTree\ t\ c)\ cs)$
 $up\ _ = Nothing$
 $right\ (TZ\ l\ (NodeL\ ()\ r : cs)) = Just\ (TZ\ r\ (NodeR\ l\ () : cs))$
 $right\ _ = Nothing$

The function *right* fails if there is no immediate right sibling. If we want to move to the *right* even if there is no immediate sibling, we can define

$right' :: TreeZipper\ a \rightarrow Maybe\ (TreeZipper\ a)$
 $right'\ z = right\ z\ 'mplus'\ (up\ z\ \gg\ right'\ \gg\ down)$

3 (6 points). Assuming a suitable instance ○

instance *Arbitrary* *a* \Rightarrow *Arbitrary* (*TreeZipper* *a*)

consider the QuickCheck property

$downUp :: (Eq\ a) \Rightarrow TreeZipper\ a \rightarrow Bool$
 $downUp\ z = (down\ z\ \gg\ up) == Just\ z$

Give a counterexample for this property, and suggest how the property can be improved so that the test will pass. •

4 (4 points). Is

$$\begin{aligned} \text{left} &:: \text{TreeZipper } a \rightarrow \text{Maybe } (\text{TreeZipper } a) \\ \text{left } z &= \text{up } z \gg \text{down} \end{aligned}$$

a suitable definition for *left*? Give reasons for your answer. [No more than 30 words.]

•

Solution 4. It moves to the left sibling when possible. In the root, *left* fails (which is fine). In other nodes without left siblings, *left* returns to the same place.

○

5 (6 points). The concept of a *one-hole context* is not limited to binary trees. Give a suitable definition of *ListCtx* such that we can define

$$\text{data ListZipper } a = \text{LZ } [a] [\text{ListCtx } a]$$

and in principle play the same game as with the zipper for trees. Also define the function

$$\text{plugList} :: [a] \rightarrow \text{ListCtx } a \rightarrow [a]$$

the combines a list context with a list.

•

Solution 5. There is no way to descend into an empty list, and only one way to descend into a non-empty list. When descending to the tail, we have to remember the element we pass, so the list context contains a single element:

$$\text{type ListCtx } a = a$$

Plugging is just cons-ing:

$$\text{plugList} = \text{flip } (:)$$

○

6 (6 points). Discuss the necessity of *up*, *down*, *left* and *right* functions for the *ListZipper*, and describe what they would do. No need to define them (although it is ok to do so). [No more than 40 words.]

•

Solution 6. The functions *up* and *down* correspond to moving left and right in the list, respectively:

$$\begin{aligned} \text{up} &:: \text{ListZipper } a \rightarrow \text{Maybe } (\text{ListZipper } a) \\ \text{up } (\text{LZ } xs \ (c : cs)) &= \text{Just } (\text{LZ } (c : xs) \ cs) \\ \text{up } _ &= \text{Nothing} \\ \text{down} &:: \text{ListZipper } a \rightarrow \text{Maybe } (\text{ListZipper } a) \\ \text{down } (\text{LZ } (x : xs) \ cs) &= \text{Just } (\text{LZ } xs \ (x : cs)) \\ \text{down } _ &= \text{Nothing} \end{aligned}$$

The functions *left* and *right* are not needed, as there are no siblings in the case of lists.

○

Type isomorphisms (12 points total)

7 (6 points). A different definition for one-hole contexts of trees is the following:

```
data Dir = L | R
type TreeCtx' a = (Dir, Tree a)
```

Show that, ignoring undefined values, the types *TreeCtx* and *TreeCtx'* are isomorphic, by giving conversion functions and stating the properties that the conversion functions must adhere to (*no proofs required*). •

Solution 7. The conversion functions are:

```
from :: TreeCtx a → TreeCtx' a
from (NodeL () r) = (L, r)
from (NodeR l ()) = (R, l)
to :: TreeCtx' a → TreeCtx a
to (L, r) = NodeL () r
to (R, l) = NodeR l ()
```

The conversion functions must be mutual inverses:

```
∀(c :: TreeCtx a).  to (from c) ≡ c
∀(c :: TreeCtx' a). from (to c) ≡ c
```

It is very easy to see that these properties hold. ○

8 (6 points). In Haskell's lazy setting, how many different values are there of type *TreeCtx Bool* if we restrict the occurrences of *Tree Bool* to be leaves. And how many different values are there of type *TreeCtx' Bool* given the same restriction? (Hint: note that the use of *()* in the definition of *TreeCtx* is relevant here.) •

Solution 8. For *TreeCtx Bool* there are thirteen (or seventeen):

- \perp ,
- for *NodeL*, there are six:


```
NodeL  $\perp$  (Leaf  $\perp$ ), NodeL () (Leaf  $\perp$ ), NodeL  $\perp$  (Leaf True), NodeL  $\perp$  (Leaf False),
NodeL () True, NodeL () False,
```
- and analogously, we get six for *NodeR*.

It is also ok to count *NodeL $\perp \perp$* , *NodeL () \perp* , *NodeR $\perp \perp$* and *NodeR () \perp* .

For *TreeCtx' Bool* there are ten (or thirteen): \perp , $(\perp, \text{Leaf } \perp)$, $(L, \text{Leaf } \perp)$, $(R, \text{Leaf } \perp)$, $(\perp, \text{Leaf True})$, $(\perp, \text{Leaf False})$, $(L, \text{Leaf True})$, $(L, \text{Leaf False})$, $(R, \text{Leaf True})$, $(R, \text{Leaf False})$. If you counted the extra values before, then we should count (\perp, \perp) , (L, \perp) , and (R, \perp) here as well. ○

Lenses (14 points total, plus 5 bonus points)

A so-called *lens* is (among other things) a way to access a substructure of a larger structure by grouping a function to extract the substructure with a function to update the substructure:

$$\mathbf{data} \ a \mapsto b = \mathit{Lens} \ \{ \mathit{extract} :: a \rightarrow b, \\ \mathit{insert} :: b \rightarrow a \rightarrow a \}$$

(We assume here that we enable infix type constructors, and that \mapsto is a valid symbol for such a constructor.)

Lenses are supposed to adhere to the following two *extract/insert* laws:

$$\begin{aligned} \forall (f :: a \mapsto b) \ (x :: a). \quad & \mathit{insert} \ f \ (\mathit{extract} \ f \ x) \ x \equiv x \\ \forall (f :: a \mapsto b) \ (x :: b) \ (y :: a). \quad & \mathit{extract} \ f \ (\mathit{insert} \ f \ x \ y) \equiv x \end{aligned}$$

A trivial lens is the identity lens that returns the complete structure:

$$\begin{aligned} \mathit{idLens} &:: a \mapsto a \\ \mathit{idLens} &= \mathit{Lens} \ \{ \mathit{extract} = \mathit{id}, \mathit{insert} = \mathit{const} \} \end{aligned}$$

It is trivial to see that *idLens* fulfills the two laws.

9 (4 points). Define a lens that accesses the focus component of a tree zipper structure:

$$\mathit{focus} :: \mathit{TreeZipper} \ a \mapsto \mathit{Tree} \ a$$

•

Solution 9.

$$\begin{aligned} \mathit{focus} &= \mathit{Lens} \ \{ \mathit{extract} = \lambda (TZ \ t \ cs) \rightarrow t, \\ &\quad \mathit{insert} = \lambda t \ (TZ \ _ \ cs) \rightarrow TZ \ t \ cs \} \end{aligned}$$

○

10 (4 points). Define a function that updates the substructure accessed by a lens according to the given function:

$$\mathit{update} :: (a \mapsto b) \rightarrow (b \rightarrow b) \rightarrow (a \rightarrow a)$$

•

Solution 10.

$$\mathit{update} \ (\mathit{Lens} \ \mathit{ext} \ \mathit{ins}) \ f \ x = \mathit{ins} \ (f \ (\mathit{ext} \ x)) \ x$$

○

Lenses can be composed. Structures that support identity and composition are captured by the following type class:

```
class Category cat where
  id  :: cat a a
  (◦) :: cat b c → cat a b → cat a c
```

For instance, functions are an instance of the category class, with the usual definitions of identity and function composition:

```
instance Category (→) where
  id  = Prelude.id
  (◦) = (Prelude.◦)
```

11 (6 points). Define an instance of the *Category* class for lenses:

```
instance Category (↔) where
  ...
```

Solution 11.

```
instance Category (↔) where
  id = idLens
  (◦) f g =
    Lens { extract = extract f ◦ extract g,
          insert  = update g ◦ insert f }
```

12 (5 bonus points). Prove using equational reasoning that if the two *extract/insert* laws stated above hold for both *f* and *g*, then they also hold for *f* ◦ *g*.

Solution 12. Let $x :: a, f :: b \mapsto c, g :: a \mapsto b$.

$$\begin{aligned}
& \text{insert } (f \circ g) (\text{extract } (f \circ g) x) x \\
\equiv & \quad \{ \text{definition of insert} \} \\
& (\text{update } g \circ \text{insert } f) (\text{extract } (f \circ g) x) x \\
\equiv & \quad \{ \text{definition of } (\circ) \} \\
& \text{update } g (\text{insert } f (\text{extract } (f \circ g) x)) x \\
\equiv & \quad \{ \text{definition of update} \} \\
& \text{insert } g (\text{insert } f (\text{extract } (f \circ g) x) (\text{extract } g x)) x \\
\equiv & \quad \{ \text{definition of extract} \} \\
& \text{insert } g (\text{insert } f ((\text{extract } f \circ \text{extract } g) x) (\text{extract } g x)) x
\end{aligned}$$

$$\begin{aligned}
&\equiv \{ \text{definition of } (\circ) \} \\
&\quad \text{insert } g \text{ (insert } f \text{ (extract } f \text{ (extract } g \text{ } x)) \text{ (extract } g \text{ } x)) } x \\
&\equiv \{ \text{assumption on } f \} \\
&\quad \text{insert } g \text{ (extract } g \text{ } x) x \\
&\equiv \{ \text{assumption on } g \} \\
&\quad x
\end{aligned}$$

Now let $x :: b, y :: a, f :: b \mapsto c$ and $g :: a \mapsto b$.

$$\begin{aligned}
&\text{extract } (f \circ g) \text{ (insert } (f \circ g) \text{ } x \text{ } y) \\
&\equiv \{ \text{definition of } \text{extract} \} \\
&\quad (\text{extract } f \circ \text{extract } g) \text{ (insert } (f \circ g) \text{ } x \text{ } y) \\
&\equiv \{ \text{definition of } (\circ) \} \\
&\quad \text{extract } f \text{ (extract } g \text{ (insert } (f \circ g) \text{ } x \text{ } y)) \\
&\equiv \{ \text{definition of } \text{insert} \} \\
&\quad \text{extract } f \text{ (extract } g \text{ ((update } g \circ \text{insert } f) \text{ } x \text{ } y)) \\
&\equiv \{ \text{definition of } (\circ) \} \\
&\quad \text{extract } f \text{ (extract } g \text{ (update } g \text{ (insert } f \text{ } x) \text{ } y)) \\
&\equiv \{ \text{definition of } \text{update} \} \\
&\quad \text{extract } f \text{ (extract } g \text{ (insert } g \text{ (insert } f \text{ } x \text{ } y) \text{ } y)) \\
&\equiv \{ \text{assumption on } g \} \\
&\quad \text{extract } f \text{ (insert } f \text{ } x \text{ } y) \\
&\equiv \{ \text{assumption on } f \} \\
&\quad x
\end{aligned}$$

○

Monad transformers (22 points total)

Consider the monad *TraverseTree*, defined as follows:

type *TraverseTree* *a* = *StateT* (*TreeZipper* *a*) *Maybe*

13 (3 points). What is the kind of *TraverseTree*?

•

Solution 13.

$* \rightarrow * \rightarrow *$

○

14 (6 points). Define a function

$nav :: (TreeZipper\ a \rightarrow Maybe\ (TreeZipper\ a)) \rightarrow TraverseTree\ a\ ()$

that turns a navigation function like *down*, *up*, or *right* into a monadic operation on *TraverseTree*. •

Solution 14.

```
nav f =  
  do  
    l ← get  
    x ← lift $ f l  
    put x
```

Note that using *modify* is problematic, because we cannot lift the argument to *modify* into the outer monad. ○

Given a lense and the *MonadState* interface, we can define useful helpers to access parts of the monadic state:

```
getLens :: MonadState s m => (s ↦ a) → m a  
getLens f = gets (extract f)  
putLens :: MonadState s m => (s ↦ a) → a → m ()  
putLens f x = modify (insert f x)  
modifyLens :: MonadState s m => (s ↦ a) → (a → a) → m ()  
modifyLens f g = modify (update f g)
```

We can now define the following piece of code:

```
ops :: TraverseTree Char ()  
ops =  
  do  
    nav down  
    x ← getLens focus  
    nav right  
    putLens focus x  
    nav down  
    modifyLens focus (const $ Leaf 'x')
```

15 (6 points). Given all the functions so far and once again tree

```
tree = Node (Node (Leaf 'a') (Leaf 'b'))  
          (Node (Leaf 'c') (Leaf 'd'))
```

what is the result of evaluating the following declaration:

```
test = leave (snd (fromJust (runStateT ops (enter tree))))
```

•

Solution 15. The result is

`Node (Node (Leaf 'a') (Leaf 'b')) (Node (Leaf 'x') (Leaf 'b'))`

○

16 (7 points). Explain how a compiler based on passing dictionaries for type classes can construct the dictionary to pass to the *modifyLens* call in the last line of the definition of *ops* above. ●

Solution 16. The call to *modifyLens* requires an instance

`MonadState (TreeZipper Char) (TraverseTree Char)`

which after expanding the type synonym means

`MonadState (TreeZipper Char) (StateT (TreeZipper Char) Maybe)`

Reading classes as dictionary types, we thus need a dictionary of the type above. We have the instances

instance *Monad Maybe*
instance *Monad m* \Rightarrow *MonadState s (StateT s m)*

available, in other words, we can assume dictionaries:

`monadMaybe :: Monad Maybe`
`monadState :: Monad m \rightarrow MonadState s (StateT s m)`

The desired dictionary can thus be constructed by using

`monadState monadMaybe`

○

Trees, shapes and pointers in Agda (19 points total)

Consider the definitions of *List*, *N*, *Vec* and *Fin* in Agda. These four types are related as follows:

Natural numbers describe the *shapes* of lists (if we instantiate the element type of lists to the unit type, we obtain a type isomorphic to the natural numbers). Indexing lists by their shapes yields vectors. Finally, *Fin* is the type of *pointers* into vectors such that we can define a safe lookup function.

Now consider binary trees (as before), given in Agda by:

data *Tree* (*A* : *Set*) : *Set* **where**
 leaf : *A* \rightarrow *Tree A*
 node : *Tree A* \rightarrow *Tree A* \rightarrow *Tree A*

The type of shapes for trees is given by:

```
data Shape : Set where
  end  : Shape
  split : Shape → Shape → Shape
```

17 (5 points). Define a datatype *STree* of shape-indexed binary trees (i. e., *STree* corresponds to *Vec*):

```
data STree (A : Set) : Shape → Set where
  ...
```

•

Solution 17.

```
data STree (A : Set) : Shape → Set where
  leaf  : A → STree A end
  node  : ∀ {s t} → STree A s → STree A t → STree A (split s t)
```

○

18 (6 points). Define a datatype *Path* of shape-indexed pointers (i. e., *Path* corresponds to *Fin*):

```
data Path : Shape → Set where
  ...
```

Note that a value *p* of type *Path s* should point to an element in a tree of shape *s*.

•

Solution 18.

```
data Path : Shape → Set where
  here  : Path end
  left  : ∀ {s t} → Path s → Path (split s t)
  right : ∀ {s t} → Path t → Path (split s t)
```

○

19 (4 points). Define a function *zipWith* on shape-indexed trees that merges two trees of the same shape and combines the elements according to the given function.

```
zipWith : ∀ {A B C s} → (A → B → C) →
  STree A s → STree B s → STree C s
```

•

Solution 19.

$$\begin{aligned} \text{zipWith } f \text{ (leaf } x) \text{ (leaf } y) &= \text{leaf } (f \ x \ y) \\ \text{zipWith } f \text{ (node } l_1 \ r_1) \text{ (node } l_2 \ r_2) &= \text{node } (\text{zipWith } f \ l_1 \ l_2) \ (\text{zipWith } f \ r_1 \ r_2) \end{aligned}$$

○

20 (4 points). Define a function *lookup* on shape-indexed trees

$$\text{lookup} : \forall \{A\ s\} \rightarrow \text{STree } A \ s \rightarrow \text{Path } s \rightarrow A$$

that returns the element stored at the given path.

●

Solution 20.

$$\begin{aligned} \text{lookup } (\text{leaf } x) \text{ here} &= x \\ \text{lookup } (\text{node } l \ r) \text{ (left } p) &= \text{lookup } l \ p \\ \text{lookup } (\text{node } l \ r) \text{ (right } p) &= \text{lookup } r \ p \end{aligned}$$

○