

INFOB3TC – Solutions for the Exam

Andres Löh

Monday, 7 December 2009, 09:00–12:00

Please keep in mind that often, there are many possible solutions, and that these example solutions may contain mistakes.

Context-free grammars

1 (10 points). Let $A = \{x, y, z\}$. Give context-free grammars for the following languages over the alphabet A :

- (a) $L_1 = \{w \mid w \in A^*, \#(x, w) \geq 3\}$
- (b) $L_2 = \{w \mid w \in A^*, \#(x, w) < 3\}$
- (c) $L_1 \cap L_2$

Here, $\#(c, w)$ denotes the number of occurrences of a terminal c in a word w . •

Solution 1.

- (a) Without abbreviations:

$$\begin{aligned} S &\rightarrow C x C x C x C \\ C &\rightarrow \varepsilon \mid X C \\ X &\rightarrow x \mid y \mid z \end{aligned}$$

With EBNF-abbreviations:

$$\begin{aligned} S &\rightarrow X^* x X^* x X^* x X^* \\ X &\rightarrow x \mid y \mid z \end{aligned}$$

- (b) Without abbreviations:

$$\begin{aligned} S &\rightarrow C \mid C x C \mid C x C x C \\ C &\rightarrow \varepsilon \mid Y C \\ Y &\rightarrow y \mid z \end{aligned}$$

With EBNF-abbreviations:

$$S \rightarrow Y^* x? Y^* x? Y^*$$

$$Y \rightarrow y \mid z$$

- (c) The intersection of the two languages is the empty language. A grammar for the empty language is, for example, the empty grammar (no productions) or

$$S \rightarrow S$$

where no derivations of terminal strings from the start symbol are possible.

○

Grammar analysis and transformation

Consider the following context-free grammar G over the alphabet $\{a, b, c\}$ with start symbol S :

$$S \rightarrow SaSa$$

$$S \rightarrow SaSbSa$$

$$S \rightarrow b$$

2 (10 points). For each of the following words, answer the question whether it is in $L(G)$. If yes, give a parse tree. If not, argue informally why the word cannot be in the language.


(a) babababba

(b) bababababa

●

Solution 2.

(a) The word babababba is in $L(G)$, as can be witnessed by the following parse tree:



- (b) The word bababababa is not in $L(G)$. It contains an odd number of as , but it is easy to prove via induction on the derivations in grammar G that any word derived from G must have an even number of as .

It is also possible to argue via the length of the word (length 10 is only possible by applying the first production three times, and then derive b for all remaining occurrences of S; one can argue that this does not produce the desired word) or by analyzing derivation sequences and argue that all sufficiently long words must either contain the substring aa or bb. However, these arguments are more complicated than going via the number of as and it is easier to forget a case.

○

3 (11 points). Simplify the grammar G by transforming it in steps. Perform as many as possible of the following transformations: removal of left recursion, left factoring, and removal of unreachable productions. ●

Solution 3. This is the original grammar:

$$\begin{aligned} S &\rightarrow SaSa \\ S &\rightarrow SaSbSa \\ S &\rightarrow b \end{aligned}$$

We can first left-factor the grammar:

$$\begin{aligned} S &\rightarrow SaSR \\ S &\rightarrow b \\ R &\rightarrow a \\ R &\rightarrow bSa \end{aligned}$$

Now, we remove left recursion:

$$\begin{aligned} S &\rightarrow bZ? \\ Z &\rightarrow aSRZ? \\ R &\rightarrow a \\ R &\rightarrow bSa \end{aligned}$$

Of course, it is possible to remove left recursion first and perform left factoring later. We then get

$$\begin{aligned} S &\rightarrow bZ? \\ Z &\rightarrow aSaZ? \mid aSbSaZ? \end{aligned}$$

after removal of left recursion, and left factoring then yields:

$$\begin{aligned} S &\rightarrow bZ? \\ Z &\rightarrow aSR \\ R &\rightarrow aZ? \mid bSaZ? \end{aligned}$$

If desired, it is possible to perform more operations, but that was not expected.

○

Alternative definitions of parser combinators

In the following tasks, you are not supposed to make use of the internal implementation of parser combinators.

4 (4 points). Define $(\langle \$ \rangle)$ in terms of *succeed* and $(\langle * \rangle)$. •

Solution 4. The map function on parsers can be defined as a derived combinator simply as follows:

$$\begin{aligned}(\langle \$ \rangle) &:: (a \rightarrow b) \rightarrow \text{Parser } s \ a \rightarrow \text{Parser } s \ b \\ f \langle \$ \rangle p &= \text{succeed } f \langle * \rangle p\end{aligned}$$

○

5 (5 points). Let

$$\text{anySymbol} :: \text{Parser } s \ s$$

be a parser that consumes any single symbol in the input and returns it. The parser only fails if the end of the input has been reached. Define

$$\text{symbol} :: \text{Eq } s \Rightarrow s \rightarrow \text{Parser } s \ s$$

in terms of *anySymbol*, *succeed*, $(\gg=)$ and *empty*. •

Solution 5.

$$\text{symbol } x = \text{anySymbol} \gg= \lambda y \rightarrow \text{if } x == y \text{ then } \text{succeed } y \text{ else } \text{empty}$$

Of course, we can write *return* instead of *succeed*. We can even use **do**-notation:

$$\begin{aligned}\text{symbol } x &= \text{do} \\ &\quad y \leftarrow \text{anySymbol} \\ &\quad \text{if } x == y \text{ then } \text{return } y \text{ else } \text{empty}\end{aligned}$$

Although returning *y* is preferable over returning *x* (exercise: why?), returning *x* is ok as a solution. ○

Combinators for permutations

6 (4 points). Write a parser combinator

$$\text{perms2} :: \text{Parser } s \ a \rightarrow \text{Parser } s \ b \rightarrow \text{Parser } s \ (a, b)$$

such that *perms2 p q* parses *p* followed by *q*, or *q* followed by *p*, and returns the results in a pair. Pay attention to the order in which the results are returned! •

Solution 6. Written such that the symmetry becomes most obvious:

$$\begin{aligned} perms2\ p\ q &= (\lambda x\ y \rightarrow (x, y)) <\$> p <*> q \\ &<|> (\lambda y\ x \rightarrow (x, y)) <\$> q <*> p \end{aligned}$$

The main difficulty is that we have to reorder the results so that the types match.

○

7 (10 points). Now write a parser combinator

$$perms3 :: Parser\ s\ a \rightarrow Parser\ s\ b \rightarrow Parser\ s\ c \rightarrow Parser\ s\ (a, b, c)$$

where $perms3\ p\ q\ r$ parses any permutation of p , q and r .

If you find a way of improving the efficiency of the resulting parser, explain (for example, in terms of the underlying grammar) what has to be done. It is not necessary to give the resulting parser, however.

●

Solution 7. The first approach is probably the following:

$$\begin{aligned} perms3\ p\ q\ r &= (\lambda x\ y\ z \rightarrow (x, y, z)) <\$> p <*> q <*> r \\ &<|> (\lambda x\ z\ y \rightarrow (x, y, z)) <\$> p <*> r <*> q \\ &<|> (\lambda y\ x\ z \rightarrow (x, y, z)) <\$> q <*> p <*> r \\ &<|> (\lambda y\ z\ x \rightarrow (x, y, z)) <\$> q <*> r <*> p \\ &<|> (\lambda z\ x\ y \rightarrow (x, y, z)) <\$> r <*> p <*> q \\ &<|> (\lambda z\ y\ x \rightarrow (x, y, z)) <\$> r <*> q <*> p \end{aligned}$$

However, this parser is in clear need for left-factoring. The grammar corresponding to the parser above is:

$$\begin{array}{l} S \rightarrow P\ Q\ R \\ \quad | \ P\ R\ Q \\ \quad | \ Q\ P\ R \\ \quad | \ Q\ R\ P \\ \quad | \ R\ P\ Q \\ \quad | \ R\ Q\ P \end{array}$$

which can be left-factored to

$$\begin{aligned} S &\rightarrow P\ X\ | \ Q\ Y\ | \ R\ Z \\ X &\rightarrow Q\ R\ | \ R\ Q \\ Y &\rightarrow P\ R\ | \ R\ P \\ Z &\rightarrow P\ Q\ | \ Q\ P \end{aligned}$$

If someone wrote this, it was sufficient. But now, X , Y , and Z are permutations of two elements, so it is relatively easy to write the parser in an efficient, left-factored way:

$$\begin{aligned} perms3\ p\ q\ r &= (\lambda x\ (y, z) \rightarrow (x, y, z)) <\$> p <*> perms2\ q\ r \\ &<|> (\lambda y\ (x, z) \rightarrow (x, y, z)) <\$> q <*> perms2\ p\ r \\ &<|> (\lambda z\ (x, y) \rightarrow (x, y, z)) <\$> r <*> perms2\ p\ q \end{aligned}$$

○

Parsing logical propositions

Here is a grammar for logical propositions with start symbol P :

$$\begin{array}{l}
 P \rightarrow P \wedge P \\
 \quad | \quad P \vee P \\
 \quad | \quad P \Rightarrow P \\
 \quad | \quad \neg P \\
 \quad | \quad \textit{Ident} \\
 \quad | \quad (P) \\
 \quad | \quad 1 \\
 \quad | \quad 0
 \end{array}$$

Propositions can be composed from the constants true (1) and false (0) by using negation, conjunction, disjunction and implication, and parentheses for grouping.

Furthermore, propositions can contain variables – the nonterminal *Ident* represents an identifier consisting of one or more letters.

A corresponding abstract syntax in Haskell is:

```

data P = And    P P
      | Or      P P
      | Implies P P
      | Not     P
      | Var     String
      | Const   Bool

```

8 (10 points). Resolve the operator priorities in the grammar as follows: negation (\neg) binds stronger than implication (\Rightarrow), which in turn binds stronger than conjunction (\wedge), which in turn binds stronger than disjunction (\vee). Furthermore, implication associates to the right, whereas conjunction and disjunction associate to the left. Give the resulting grammar. ●

Solution 8. We split P into several nonterminals, corresponding to the different priority levels:

$$\begin{array}{l}
 P \rightarrow P_1 \\
 P_1 \rightarrow P_1 \vee P_2 \mid P_2 \\
 P_2 \rightarrow P_2 \wedge P_3 \mid P_3 \\
 P_3 \rightarrow P_4 \Rightarrow P_3 \mid P_4 \\
 P_4 \rightarrow \neg P_4 \mid \textit{Ident} \mid (P) \mid 1 \mid 0
 \end{array}$$

There should not be any surprises in the resulting grammar. I think it's actually unusual to have implication bind as strong. After placing the exam it occurred to me that it probably makes more sense to have it bind weakest. But then again, it does not make any difference for the difficulty of the assignment. ○

9 (11 points). Give a parser that recognizes the grammar from Task 8 and produces a value of type P :

$parseP :: Parser Char P$

You can assume that the symbols \neg , \Rightarrow , \wedge , and \vee are just characters. You can use *chainl* and *chainr*, but if you want more advanced abstractions such as *gen* from the lecture notes, you have to define them yourself. You may assume that spaces are not allowed in the input. •

Solution 9. This is a rather direct transcription using *chainl* and *chainr*:

```

parseP = p1
p1     = chainl p2 (Or    <$ symbol '∨' )
p2     = chainl p3 (And   <$ symbol '∧' )
p3     = chainr p4 (Implies <$ symbol '⇒' )
p4     = Not <$ symbol '¬' <*> p4
        <|> Var <$> some (satisfy isLetter)
        <|> parenthesised parseP
        <|> Const True  <$ symbol '1'
        <|> Const False <$ symbol '0'

```

Using *identifier*, *many₁* or *greedy₁* for the *Var* case is also ok. ○

10 (10 points). Define an algebra type and a fold function for type P . •

Solution 10. We just apply the systematic translation:

```

type PAlgebra r = ( r → r → r,    — And
                    r → r → r,    — Or
                    r → r → r,    — Implies
                    r → r,        — Not
                    String → r,   — Var
                    Bool → r)     — Const

foldP :: PAlgebra r → P → r
foldP (and, or, implies, not, var, const) = f
  where f (And    x1 x2) = and    (f x1) (f x2)
        f (Or     x1 x2) = or     (f x1) (f x2)
        f (Implies x1 x2) = implies (f x1) (f x2)
        f (Not     x   ) = not    (f x)
        f (Var     x   ) = var    x
        f (Const   x   ) = const  x

```

No surprises here. ○

11 (10 points). Using the algebra and fold (or alternatively directly), define an evaluator for propositions:

$evalP :: P \rightarrow Env \rightarrow Bool$

The environment of type Env should map free variables to Boolean values. You can either use a list of pairs or a finite map with the following interface to represent the environment:

data $Map\ k\ v$ — abstract type, maps keys of type k to values of type v

$empty :: Map\ k\ v$

$(!) :: Ord\ k \Rightarrow Map\ k\ v \rightarrow k \rightarrow v$

$insert :: Ord\ k \Rightarrow k \rightarrow v \rightarrow Map\ k\ v \rightarrow Map\ k\ v$

$delete :: Ord\ k \Rightarrow k \rightarrow Map\ k\ v \rightarrow Map\ k\ v$

$member :: Ord\ k \Rightarrow k \rightarrow Map\ k\ v \rightarrow Bool$

$fromList :: Ord\ k \Rightarrow [(k, v)] \rightarrow Map\ k\ v$

•

Solution 11. We assume

type $Env = Map\ String\ Bool$

The algebra is similar to the evaluator for expressions discussed in the lectures:

$evalAlgebra :: PAlgebra\ (Env \rightarrow Bool)$

$evalAlgebra = (\lambda x_1\ x_2\ e \rightarrow x_1\ e \ \&\&\ x_2\ e,$
 $\lambda x_1\ x_2\ e \rightarrow x_1\ e \ \|\ x_2\ e,$
 $\lambda x_1\ x_2\ e \rightarrow not\ (x_1\ e) \ \|\ x_2\ e,$
 $\lambda x\ e \rightarrow not\ (x\ e),$
 $\lambda x\ e \rightarrow e!\ x,$
 $\lambda x\ e \rightarrow x)$

$evalP = foldP\ evalAlgebra$

The environment never changes, so defining

$evalAlgebra :: Env \rightarrow PAlgebra\ Bool$

$evalAlgebra\ e = ((\&\&), (\|), (\lambda x_1\ x_2 \rightarrow not\ x_1 \ \|\ x_2), not, (e!), id)$

is simpler and ok as well.

○

12 (5 points). Implement a tautology checker for propositions of type P :

$tautology :: P \rightarrow Bool$

A proposition is a tautology if and only if it evaluates to *True* regardless of the values of any of its free variables.

It may be helpful to use the following function *assignments* that produces a list of all possible Boolean assignments for a list of identifiers:

```
assignments :: [String] → [[(String, Bool)]]
assignments [] = [[]]
assignments (n : ns) = [(n, x) : xs | x ← [True, False], xs ← assignments ns]
```

You can use *evalP* – even if you have not implemented it – in the definition of *tautology*. •

Solution 12. We need a way to discover the free variables in *P*:

```
freeVarsAlgebra :: PAlgebra [String]
freeVarsAlgebra = ((++), (++), (++), id, (:[]), const [])
freeVars :: P → [String]
freeVars = foldP freeVarsAlgebra
```

This algebra simply collects all the variables. Now we can define the tautology checker:

```
tautology p = all (λe → evalP p (fromList e)) (assignments (freeVars p))
```

For all the assignments corresponding to the free variables, the proposition has to evaluate to true. The function *all* is a standard function from the prelude and can be defined as follows:

```
all :: (a → Bool) → [a] → Bool
all p = and . map p
and :: [Bool] → Bool
and = foldr (&&) True
```

○

13 (meta question). How many out of the 100 possible points do you think you will get for this exam? •

Solution 13. I ask this question because it is the first time I’m setting an exam for the course, and I’m genuinely interested how difficult the students perceive the exam to be, and how good the students are in judging their own performance. Obviously, the answer to this question has no relevance for the final result. ○