

Only hand in the sheet with the answers, while showing your legitimation. Don't forget to fill out your name! Do not forget necessary parentheses!

1 The function *intersperse*

Write a function *intersperse* :: $a \rightarrow [a] \rightarrow [a]$, which places its first argument between the elements of its second argument; i.e. **intersperse** 'a' "xyz" should return "xayaz". (2 points)

There are quite a few possible solutions. The most straightforward one is e.g.:

```
intersperse a (x : y : ys) = x : a : intersperse a (y : ys)
intersperse _ xs           = xs
```

The second case takes care of the empty *xs* and an *xs* of length 1. Unfortunately, as I explained in the last lecture, this function is not as productive as it could be.

Alternative solutions are:

```
intersperse a xs = tail ( foldr (\x r -> a : x : r) [] xs)
intersperse a   = tail . foldr (\x r -> a : x : r) []
intersperse a   = tail . foldr (\x -> (a:) . (x:)) []
intersperse a   = tail . foldr ((a:) .) . (:) []
intersperse a   = tail . concat . map (\x -> [a, x])
```

2 Type inference

1. What is the type of *map map* (1 point)
2. Explain in at most 40 words why *foldl map* is not well-typed (1 point).

Let us assume that the first *map* has type: $(a \rightarrow b) \rightarrow [a] \rightarrow [b]$ and the second one $(c \rightarrow d) \rightarrow [c] \rightarrow [d]$, which has to match with $(a \rightarrow b)$. Hence we infer that $a = c \rightarrow d$ and $b = [c] \rightarrow [d]$; we conclude that the result type is thus $[a] \rightarrow [b]$ with the above substitution: $[[c \rightarrow d] \rightarrow [[c] \rightarrow [d]]]$.

We know that *foldl* :: $(b \rightarrow a \rightarrow b) \rightarrow b \rightarrow [a] \rightarrow b$, and that *map* :: $(c \rightarrow d) \rightarrow [c] \rightarrow [d]$ which has to line up with the $b \rightarrow a \rightarrow b$. We conclude both $b = c \rightarrow d$ and $b = [d]$, which cannot both be the case.

3 The function *foldl*

1. What is the type of the expression *foldl* (*flip* (:)) [], where *flip* $f\ x\ y = f\ y\ x$ (1 point). Hint: first write down the types of (:), *foldl* and *flip* on your piece of scrap paper.
2. Give a definition of the function *reverse* :: $[a] \rightarrow [a]$ which uses an *accumulating parameter* in order to avoid inefficiencies.

```
(:) :: a -> [a] -> [a]
flip :: (a -> b -> c) -> (b -> a -> c)
flip (:) :: [a] -> a -> [a]
foldl :: (b -> a -> b) -> b -> [a] -> b
foldl (flip (:)) [] :: [a] -> [a]
```

Actually this expression is equivalent to the function *reverse*:

```
reverse xs = reverse' xs []
  where reverse' [] r = r
        reverse' (x : xs) r = reverse' xs (x : r)
```

which you can get by substituting the arguments (*flip* (:) and [] in the definition of *foldl*.

Many, many of you have written something like:

```
reverse [] = []  
reverse (x : xs) = reverse xs : x
```

This does not use an accumulating parameter, besides that you cannot use : to put an extra element at the end of the list (use *reverse xs ++ [a]* instead), and the parentheses are missing, etc.

4 Input and output

Write a function *ask* :: [String] → IO [String] which takes a list of questions (each a *String*), poses those question on the terminal, reads an answer (assume 'y' or 'n' as the first character in the answer line) and *return*-s only those questions to which a positive answer was given (2 points).

```
ask [] = return []  
ask (q : qs) = do putStrln q  
                  (a: _) ← getLine  
                  as ← ask qs  
                  if a == 'y' then return (q : as) else return as
```

or

```
return (if a == 'y' then (q:) else id) as)
```

5 List based functions

1. Write the function *subs* :: [a] → [[a]] which returns all sublists of the argument list (1 point).
2. Change this function into a function *subs_asc* such that it only return those lists which are ascending (stijgend) (1 point). Do not use the function *filter* since this may be very inefficient; make the filtering part of the generation process.

```
subs [] = [[]]  
subs (x : xs) = map (x:) subsxs ++ subsxs  
                where subsxs = subs xs  
  
subs_asc [] = [[]]  
subs_asc (x : xs) = [x : l | l ← subsxs, null l ∨ x ≤ head l] ++ subsxs  
                    where subsxs = subs_asc xs
```